- Is it possible to observe
  $\gamma + \gamma \to e^+ + e^-$?

  If so, How ?

- Related Processes

(1) $e^+ + e^- \to \gamma + \gamma$

a form of electron–positron annihilation.

(2) $p + p \to$ 2 jets at the LHC

One subprocess is

$$\text{gluon} + \text{gluon} \to q + \bar{q}$$

analogous to $\gamma\gamma \to e\bar{e}$. (But this has 3 Feynman diagrams of QCD.)

(3) Production of an electromagnetic shower in the atmosphere (HAWC).



etc

But something is missing from this picture;



is not possible.

γ + Nitrogen nucleus $\to e + \bar{e}$



$\gamma\gamma \to e\bar{e}$

CHAPTER 14 - APPLICATIONS : QED

OUTLINE of the chapter





Section 14.6 ▸

$$\gamma(k_1) + \gamma(k_2) \to e^-(p_1) + e^+(p_2)$$

Kinematics in the center of mass frame of reference.

$k_1^\mu = (\ \omega\ ,\ 0\ 0\ \omega\ )$ and $k_2^\mu = (\ \omega\ ,\ 0\ \ 0\ \ -\omega\ )$

$p_1^\mu = (\ E_e\ ,\ p \sin\theta\ ,\ 0\ ,\ p \cos\theta\ )$

$p_2^\mu = (\ E_e\ ,\ -p \sin\theta\ ,\ 0\ ,\ -p \cos\theta\ )$

Conservation of energy $\omega+\omega = E_e+E_e$, implies $E_e = \omega$.

Then $p = \beta E_e$ = SQRT($E_e^2 - m^2$) ;

so $p = \beta\,\omega$ and $m^2 = \omega^2\,(1 - \beta^2)$;

or, $\omega = m\,/$SQRT($1 - \beta^2$) .

Mandelstam variables in the center of mass frame of reference:

I'm using the standard Mandelstam variables; Maiani and Benhar use different definitions for s, t, u for this example.



$$s = (k_1 + k_2)^2 = 4\omega^2 = (p_1 + p_2)^2 = 4E_e^2$$

$$t = (k_1 - p_1)^2 = (\omega - E_e)^2 - (-p\sin\theta)^2 - (\omega - p\cos\theta)^2$$

$$= -p^2 - \omega^2 + 2\omega p\cos\theta = -\omega^2(\beta^2 + 1 - 2\beta\cos\theta)$$

$$u = (k_1 - p_2)^2 = (\omega - E_e)^2 - (p\sin\theta)^2 - (\omega + p\cos\theta)^2$$

$$= -p^2 - \omega^2 - 2\omega p\cos\theta = -\omega^2(\beta^2 + 1 + 2\beta\cos\theta)$$

Note: $s + t + u = 4\omega^2 - 2\omega^2(\beta^2 + 1) = 2\omega^2(1 - \beta^2) = 2m^2$ ✔

- **Calculate the square of the transition amplitude**

There are two Feynman diagrams

$M_a =$



$$= e^2\, \bar{u}_1\, \gamma^\mu S_F(\, p_1 - k_1\,)\, \gamma^\nu\, v_2\, \varepsilon_{1\mu}\varepsilon_{2\nu}$$

$M_b =$



$$= e^2\, \bar{u}_1\, \gamma^\mu S_F(\, p_1 - k_2\,)\, \gamma^\nu\, v_2\, \varepsilon_{1\nu}\varepsilon_{2\mu}$$

$$|\,\mathbf{M_a}\,|^2 = e^4\, \bar{u}_1\, \gamma^\mu S_F(\, p_1 - k_1\,)\, \gamma^\nu\, v_2 \quad \varepsilon_{1\mu}\varepsilon_{2\nu}$$
$$\bar{v}_2\, \gamma^\beta S_F(\, p_1 - k_1\,)\, \gamma^\alpha\, u_1 \quad \varepsilon_{1\alpha}\varepsilon_{2\beta}$$

For unpolarized scattering:

$$|\,\mathbf{M_a}\,|^2 = \tfrac{1}{4}\, e^4 /(t - m^2)^2 /(2m)^2$$
$$\mathrm{Tr}\,\{\, \gamma^\mu (\not{p}_1 - \not{k}_1 + m)\, \gamma^\nu\, (\,\not{p}_2 - m)$$
$$\gamma_\nu\, (\,\not{p}_1 - \not{k}_1 + m\,)\, \gamma_\mu\, (\not{p}_1 + m)\,\}$$

$$|\,\mathbf{M_b}\,|^2 = e^4\, \bar{u}_1\, \gamma^\mu S_F(\, p_1 - k_2\,)\, \gamma^\nu\, v_2 \quad \varepsilon_{1\nu}\varepsilon_{2\mu}$$
$$\bar{v}_2\, \gamma^\beta S_F(\, p_1 - k_2\,)\, \gamma^\alpha\, u_1 \quad \varepsilon_{1\beta}\varepsilon_{2\alpha}$$

For unpolarized scattering:

$$|\,\mathbf{M_b}\,|^2 = \tfrac{1}{4}\, e^4 /(u - m^2)^2 /(2m)^2$$
$$\mathrm{Tr}\,\{\, \gamma^\mu (\not{p}_1 - \not{k}_2 + m)\, \gamma^\nu\, (\,\not{p}_2 - m)$$
$$\gamma_\nu\, (\,\not{p}_1 - \not{k}_2 + m\,))\, \gamma_\mu\, (\not{p}_1 + m)\,\}$$

Process $\gamma(k1) + \gamma(k2) \to e(p1) + \gamma(p2)$

```
In[1]:= $LoadFeynArts = False;
        << HighEnergyPhysics`FeynCalc`
```

Loading FeynCalc from /home/stump/.Mathematica/App
FeynCalc 8.2.0 For help, type ?FeynCalc, open FeynC

```
In[3]:= (* scalar products *)
        Remove["Global`*"]
        ScalarProduct[k1, k1] = 0;
        ScalarProduct[k2, k2] = 0;
        ScalarProduct[p1, p1] = m^2;
        ScalarProduct[p2, p2] = m^2;

In[8]:= (* Mandelstam variables *)
        ScalarProduct[k1, k2] = s / 2;
        ScalarProduct[p1, p2] = (s - 2 m^2) / 2;
        ScalarProduct[k1, p1] = (m^2 - t) / 2;
        ScalarProduct[p2, k2] = (m^2 - t) / 2;
        ScalarProduct[k1, p2] = (m^2 - u) / 2;
        ScalarProduct[p1, k2] = (m^2 - u) / 2;
```

Msq = square of the amplitude

term |Ma|^2

```
In[14]:= coeffAA = e^4 / (t - m^2)^2 / (16 m^2);
         traceAA = Tr[
            GA[μ].(GS[p1] - GS[k1] + m).GA[ν].(GS[p2] - m).
             GA[ν].(GS[p1] - GS[k1] + m).GA[μ].(GS[p1] + m)];
         ξAA = Expand[traceAA /. {s → 2 * m^2 - t - u}]
```

Out[16]= $-8\, m^4 - 24\, m^2\, t - 8\, m^2\, u + 8\, t\, u$

term |Mb|^2

```
In[17]:= coeffBB = e^4 / (u - m^2)^2 / (16 m^2);
         traceBB = Tr[
            GA[μ].(GS[p1] - GS[k2] + m).GA[ν].(GS[p2] - m).
             GA[ν].(GS[p1] - GS[k2] + m).GA[μ].(GS[p1] + m)];
         ξBB = Expand[traceBB /. {s → 2 * m^2 - t - u}]
         (* Note that ξUU is ξSS with the exchange s and u *)
```

Out[19]= $-8\, m^4 - 8\, m^2\, t - 24\, m^2\, u + 8\, t\, u$

The interference terms

$$M_a M_b^* = e^2\, \bar{u}_1\, \gamma^\mu S_F(p_1 - k_1)\, \gamma^\nu v_2\, \varepsilon_{1\mu}\, \varepsilon_{2\nu}$$

$$e^2\, \bar{v}_2\, \gamma^\beta S_F(p_1 - k_2)\, \gamma^\alpha u_1\, \varepsilon_{1\beta}\, \varepsilon_{2\alpha}$$

For unpolarized scattering:

$$M_a M_b^* = \tfrac{1}{4} e^4 /(t - m^2)^2 /(u - m^2) /(2m)^2$$

$$\mathrm{Tr}\,\{ \gamma^\mu (\not{p}_1 - \not{k}_1 + m)\, \gamma^\nu (\not{p}_2 - m)$$

$$\gamma_\mu (\not{p}_1 - \not{k}_2 + m)\, \gamma_\nu (\not{p}_1 + m) \}$$

Matrix element squared; the interference terms

```
In[20]:= coeffAB = e^4 / (t - m^2) / (u - m^2) / (16 m^2);
         traceAB = Tr[
             GA[μ].(GS[p1] - GS[k1] + m).GA[ν].(GS[p2] - m).
              GA[μ].(GS[p1] - GS[k2] + m).GA[ν].(GS[p1] + m)];
         ξAB = Expand[traceAB /. {s → 2 * m^2 - t - u}]
         (* Note that ξAB is real *)
```

Out[22]= $-16\, m^4 - 8\, m^2\, t - 8\, m^2\, u$

Summing Up

```
In[23]:= MsqAA = coeffAA * ξAA;
         MsqBB = coeffBB * ξBB;
         MsqINT = 2 * coeffAB * ξAB;
         Msq = MsqAA + MsqBB + MsqINT
```

Out[26]= $$\frac{e^4\,(-16\, m^4 - 8\, m^2\, t - 8\, m^2\, u)}{8\, m^2\,(t - m^2)(u - m^2)} + \frac{e^4\,(-8\, m^4 - 8\, m^2\, t - 24\, m^2\, u + 8\, t\, u)}{16\, m^2\,(u - m^2)^2} + \frac{e^4\,(-8\, m^4 - 24\, m^2\, t - 8\, m^2\, u + 8\, t\, u)}{16\, m^2\,(t - m^2)^2}$$

```
(* Compare the Final Result to Equation (14.156) *)
test = Expand[Msq /. {t → tM + m^2, u → uM + m^2}];
test = Expand[test /. {e → 1}];
test = Expand[test * m^2 /. {tM → m^2 / ρ , uM -> m^2 / η}]
compare = (ρ + η) ^2 + (ρ + η) - 1 / 4 * (ρ / η + η / ρ)
Expand[test + 2 * compare] (* Note: test = - 2 compare *)
Asq = e^4 / m^2 * (-2 * compare) /. {ρ → m^2 / tM, η → m^2 / uM};
Asq = Asq /. {e → Sqrt[4 * π * α]}
```

Out[29]= $-2\,\eta^2 - 4\,\eta\,\rho + \frac{\eta}{2\,\rho} + \frac{\rho}{2\,\eta} - 2\,\eta - 2\,\rho^2 - 2\,\rho$

Out[30]= $(\eta + \rho)^2 + \frac{1}{4}\left(-\frac{\eta}{\rho} - \frac{\rho}{\eta}\right) + \eta + \rho$

Out[31]= 0

Out[33]= $-\frac{32\,\pi^2\,\alpha^2\left(\left(\frac{m^2}{\mathrm{tM}} + \frac{m^2}{\mathrm{uM}}\right)^2 + \frac{m^2}{\mathrm{tM}} + \frac{m^2}{\mathrm{uM}} + \frac{1}{4}\left(-\frac{\mathrm{tM}}{\mathrm{uM}} - \frac{\mathrm{uM}}{\mathrm{tM}}\right)\right)}{m^2}$

Cross section in the center of mass frame of reference

In[34]:=
```
Remove[β, Δσ, Bsq, Csq]
bs = {FontFamily → Times, FontSize → 14};
```

The cross section

First recall the general formula for a 2 to 2 process

$$d\sigma = \frac{[\prod n^2]}{4 E_1 E_2 v_{rel}} (2\pi)^4 \delta^4(P_f - P_i) \frac{d^3p_3}{(2\pi)^3 2E_3} \frac{d^3p_4}{(2\pi)^3 2E_4} |M|^2 .$$

Apply it to γγ→ee in the center of mass frame of reference,

$$d\sigma = \frac{4m^2}{4\omega\sqrt{s}} \frac{1}{16\pi^2} \frac{p_3^{\,2}}{E_3E_4} \frac{E_3E_4}{\sqrt{s}\; p_3} |M|^2 \quad d\Omega_3$$

So,

$$\left.\frac{d\sigma}{d\Omega_3}\right|_{CM} = \frac{4m^2}{64\pi^2 s} \frac{p_3}{p_1} |M|^2$$

,

Here p_1 means $|\mathbf{k_1}| = \omega$;

and p_3 means $|\mathbf{p_1}| = \beta\,\omega$.

$$d\sigma = \frac{4m^2\beta(1-\beta^2)}{64\pi^2\, 4m^2} |M|^2 \, d\Omega_3$$

Also,

$$d\Omega_3 = 2\pi \; d(\cos\theta) .$$

$s = 4\,\omega^2 = 4\,E_e^{\,2} = 4\,p^2/\beta^2$; $\omega^2 = m^2(1-\beta^2)$;

In[36]:=
```
Bsq = Expand[Asq /. {tM → -2 * m^2 * (1 - β * cos) / (1 - β^2)}];
Bsq = Expand[Bsq /. {uM → -2 * m^2 * (1 + β * cos) / (1 - β^2)}];
Bsq = Simplify[Bsq]
```

Out[38]=
$$-\frac{16\,\pi^2\,\alpha^2\,(\beta^4\,(2+\cos^4-2\,\cos^2)+2\,\beta^2\,(\cos^2-1)-1)}{m^2\,(\beta\cos-1)^2\,(\beta\cos+1)^2}$$

In[39]:=
```
Δσ = 4 m^2 / (64 * π^2 * s) * β * (2 π) * Bsq;
Δσ = Δσ /. {s → 4 * m^2 / (1 - β^2)}
Δσ = Δσ /. {α → 1, m → 1};
Plot[{Δσ /. β → 0.1, Δσ /. β → 0.9}, {cos, -1, 1},
 PlotStyle → {Blue, Red}, BaseStyle → bs,
 Frame → True, FrameLabel → {"cosθ", "Δσ [arb. units]"}]
```

dσ/dΩ formula

Out[40]=
$$-\frac{\pi\,\alpha^2\,\beta\,(1-\beta^2)\,(\beta^4\,(2+\cos^4-2\,\cos^2)+2\,\beta^2\,(\cos^2-1)-1)}{2\,m^2\,(\beta\cos-1)^2\,(\beta\cos+1)^2}$$

Out[42]=



dσ /dΩ versus cos θ

red : β = 0.9

blue : β = 0.1

Total cross section for $\gamma\gamma \to e^+e^-$ versus v/c

```
In[43]:= σtot = Assuming[β > 0 && β < 1, Integrate[Δσ, {cos, -1, 1}]]
         Plot[σtot , {β, 0, 1},
          PlotStyle → Black, BaseStyle → bs, AspectRatio → 0.8,
          Frame → True, FrameLabel → {"β = v/c of the electron", "σtot
```

Out[43]= $-\frac{1}{2}\pi(\beta^2-1)\left((\beta^4-3)\log\left(\frac{2}{\beta+1}-1\right)+2\beta(\beta^2-2)\right)$ *

Out[44]=



Total cross section versus ω_γ

```
In[45]:= ParametricPlot[{1 / Sqrt[1 - β^2], σtot} , {β, 0, 0.99},
          PlotStyle → Black, BaseStyle → bs, PlotRange → {{0, 6}, {0, 2.5}},
          Frame → True, FrameLabel → {"ω/m of the photon", "σtot [arb. units
          AspectRatio → 0.8]
```

Out[45]=



* : agrees with Eq. (14.157)

The related process $e^- + e^+ \rightarrow \gamma + \gamma$,
and the mean lifetime of positronium.



Homework Problems due Friday April 14

22. Is it possible to observe $\gamma\gamma \rightarrow e^- + e^+$.
If so, how. If not, why not?

23. Calculate the mean lifetime of para-positronium. (You can use the results in the book, but fill in the gaps.)

24. Show why the annihilation process

$e^- + e^+ \rightarrow \gamma$ *cannot* occur.